

Table 1: Input for Monte Carlo simulation of the Heckman selection model

```

TITLE:          Selection modeling: Muthén-Jöreskog (1983), p. 146 with
                data generated similar to Model 1, p. 158 y missing if u=0
MONTECARLO:    NAMES = y u x;
                NOBSERVATIONS = 4000;
                NREPS = 500;
                CATEGORICAL = u; ! u = 1 if y observed
                GENERATE = u(1 p);
                MISSING = y;
MODEL POPULATION:
                x@1;
                y ON x*1;
                [y*0];
                y*1;
                f BY y*-1 u@1; ! gives residual corr = -0.5
                f@1;
                u ON x*-1;
ANALYSIS:      ESTIMATOR = MLR;
                LINK = PROBIT;
                PROCESSORS = 8;
                MCONVERGENCE = 0.00001;
                INTEGRATION = 30;
MODEL:         y ON x*1;
                [y*0];
                y*1 (v);
                f BY y*-1 (lam)
                u@1; ! gives -0.5 res. correlation
                f@1;
                u ON x*-1 (slope);
                [u$1] (thresh);
MODEL MISSING:
                ! binary y = 1 denotes missing on continuous y
                ! logit regression for y with [y] denoting intercept
                [y@15]; ! probability one of missing on y if u = 0
                y ON u@-30; ! probability zero of missing on y if u=1
MODEL CONSTRAINT:
                NEW (rescorr*-.5 probint*0 probslop*-.707107);
                rescorr = lam/(SQRT(lam*lam+v)*SQRT(1+1));
                probint = -thresh/SQRT(1+1);
                probslop = slope/SQRT(1+1);
OUTPUT:        TECH9;

```
